

Un peu de programmation impérative

Au programme officiel

Traits et éléments techniques à connaître

- Absence d'instruction ; la programmation impérative est mise en œuvre par des expressions impures ; `unit`, `()`.
- Références : type `'a ref`, notations `ref`, `!`, `:=`. Les références doivent être utilisées à bon escient.
- Séquence ;. La séquence intervient entre deux expressions.
- Boucle `while c do b done` ; boucle `for v = d to f do`

Éléments techniques devant être reconnus et utilisables après rappel

- Tableaux : type `'a array`, notations `[|...|]`, `t.(i)`, `t.(i) <- v` ; les fonctions suivantes du module `Array` : `length`, `make`, `make_matrix`, `init`, `copy`, `mem`, `exists`, `for_all`, `map` et `iter`.
- Types enregistrements immuables et mutables, notations associées.

1 Enregistrements avec champ modifiable

Dans un type enregistrement, on peut déclarer un ou plusieurs champs comme modifiable (ou *mutable*).

```
type etudiant = {  
  nom : string;  
  prenom : string;  
  mutable classe : string;  
};;  
  
let e = {  
  nom = "Tournesol";  
  prenom = "Tryphon";  
  classe = "MPSI"  
};;
```

```
type etudiant = { nom : string; prenom : string; mutable classe : string; }  
val e : etudiant = {nom = "Tournesol"; prenom = "Tryphon"; classe = "MPSI"}
```

Si le champ `c` de l'enregistrement `a` est mutable, la modification de sa valeur se fait avec `a.c <- e` où `e` est une expression.

```
e.classe <- "MP";;  
e;;
```

```
- : unit = ()  
- : etudiant = {nom = "Tournesol"; prenom = "Tryphon"; classe = "MP"}
```

Il n'y a pas d'instruction en OCaml. `e.classe <- "MP"` est en réalité une expression dont la valeur est `()` et dont le type est `unit`.

2 Références

Les variables en OCaml rencontrées jusqu'à présent ne sont pas des variables au sens traditionnel des langages de programmation, puisqu'il est impossible de modifier leur valeur. En programmation impérative, il est in-

dispensable de pouvoir utiliser des variables modifiables pour mémoriser une information évoluant au fil du programme.

En OCaml, on utilise alors une *référence* vers une valeur, c'est-à-dire une case mémoire dont on peut lire et écrire le contenu.

En pratique, on définit une référence avec le mot-clé `ref` (qui est en fait une fonction `'a -> 'a ref`).

```
let x = ref 0;;
```

```
val x : int ref = {contents = 0}
```

On remarque que son type correspond à un type enregistrement `'a ref` qui contient un unique champ de type `'a` appelé `contents`.

Le type de la valeur pointée par une référence est fixé à la création. Une référence pointant vers un objet de type `'a` a le type `'a ref`.

Pour accéder à la valeur de la référence `x`, on utilise l'*opérateur de référencement* `!` suivi du nom de la référence.

Pour modifier une référence, on utilise l'*opérateur d'affectation* `:=` précédé du nom de la référence, et suivi d'une expression.

```
x := 3;;
```

```
- : unit = ()
```

```
x := !x + 1;;
```

```
- : unit = ()
```

```
!x;;
```

```
- : int = 4
```

3 Type unit et fonctions avec effets de bord

Une fonction est dite à *effet de bord* lorsqu'elle modifie un état en dehors de son environnement local. Par extension, un opérateur est dit à *effet de bord* lorsqu'il modifie l'un de ses opérandes ; par exemple, l'opérateur d'affectation `:=` est à effet de bord.

Il est fréquent qu'une fonction à effet de bord n'ait pas besoin de renvoyer une valeur. Dans ce cas, elle renverra une valeur de type `unit`. La seule valeur de type `unit` est `()`

Remarque : Il s'agit de l'équivalent de `None` en Python (dont le type est `NoneType`).

Par exemple, les fonctions d'affichage sont des fonctions à effet de bord : `print_string`, `print_char`, `print_int`, `print_float`, `print_newline`, `print_endline`.

Une fonction qui n'a pas besoin d'un argument aura un argument de type `unit`, on la définira et on l'appellera donc avec `()`

```
let f () = print_endline "Coucou";;
```

```
val f : unit -> unit = <fun>
```

La fonction `print_endline : string -> unit` permet d'afficher une chaîne de caractères et de revenir ensuite à la ligne.

```
f ();;
```

Coucou

- : unit = ()

4 Séquences d'instructions

En Ocaml, les séquences d'instructions sont des expressions séparées par des points-virgules. L'évaluation de

$$e_1 ; e_2 ; \dots ; e_n$$

provoque les effets éventuels de ces n expressions dans l'ordre, et a la valeur de la dernière expression.

Ce dernier point implique qu'il n'y a aucun intérêt à utiliser une séquence si les expressions dont la valeur n'est pas utilisée (c'est-à-dire toutes sauf la dernière) n'ont pas d'effet de bord.

Par ailleurs, ces différentes expressions (sauf éventuellement la dernière) sont la plupart du temps de type `unit` puisque seule la valeur de la dernière expression est prise en compte.

```
let classe nom numero =  
  print_string "Vivent les ";  
  print_string nom;  
  print_int numero;  
  print_newline ();;
```

val classe : string -> int -> unit = <fun>

```
classe "MPSI" 0;;
```

Vivent les MPSI0

- : unit = ()

Lorsqu'on souhaite utiliser une séquence à la place d'une expression dans une instruction conditionnelle, il est nécessaire d'encadrer celle-ci par les mots-clés `begin` et `end` ou d'utiliser des parenthèses.

```
let est_pair n =  
  if n mod 2 = 0  
  then  
    begin  
      print_endline "nombre pair";  
      true  
    end  
  else  
    begin  
      print_endline "nombre impair";  
      false  
    end  
;;
```

val est_pair : int -> bool = <fun>

5 Boucles

5.1 Boucles inconditionnelles

Il existe deux types de boucles inconditionnelles en OCaml :

- `for indice = e1 to e2 do e3 done`
- `for indice = e1 downto e2 do e3 done`

Dans le premier cas, l'indice de boucle (de type `int`) est incrémenté de 1 en 1 entre les valeurs des expressions e_1 et e_2 , en effectuant le calcul de e_3 à chaque fois (il est donc préférable que e_3 ait un effet de bord, et pertinent que e_3 soit de type `unit`).

Le second cas est identique, mais l'indice de boucle est décrémenté.

```
for i = 1 to 10 do print_int i;  print_char ' ' done;
print_newline ();;
```

```
1 2 3 4 5 6 7 8 9 10
- : unit = ()
```

```
for i = 10 downto 1 do print_int i;  print_char ' ' done;
print_newline ();;
```

```
10 9 8 7 6 5 4 3 2 1
- : unit = ()
```

⚠ Les deux bornes sont atteintes.

5.2 Boucles conditionnelles

La syntaxe d'une boucle conditionnelle est la suivante :

- `while e1 do e2 done`

Tant que l'expression e_1 (nécessairement de type `bool`) a la valeur `true`, on évalue l'expression e_2 .

```
let a = ref 2022 in
while !a > 0 do
  print_int (!a mod 2);
  a := !a / 2
done;
print_newline ();;
```

```
01100111111
- : unit = ()
```

On lira bien évidemment l'affichage de droite à gauche.

Remarque : Si l'expression e_2 n'a pas d'effet de bord, il est peu probable que la condition puisse cesser d'être vérifiée...

6 Structure de tableaux

Les *tableaux* (ou vecteurs) sont des structures correspondant aux vecteurs mathématiques. Un tableau est une suite finie de valeurs **de même type**, modifiables, l'accès à une composante se faisant à temps constant.

La *longueur* d'un tableau (son nombre d'éléments) est fixée lors de la création et **ne peut être modifiée**

Un tableau est délimité par `[` et `]`, ses éléments sont séparés par des points-virgules.

Un tableau contenant des valeurs de type `'a'` a pour type `'a' array`.

```
let t1 = [|1;2;3;4;5;6|];;
```

```
val t1 : int array = [|1; 2; 3; 4; 5; 6|]
```

La fonction `Array.length` renvoie la longueur du tableau passé en argument.

```
Array.length t1;;
```

```
- : int = 6
```

En OCaml, les éléments d'un tableau de longueur n sont numérotés de 0 à $n - 1$.

L'accès à l'élément d'indice i du tableau t s'écrit `t.(i)`.

Cet élément peut être modifié par `t.(i) <- e` où e est une expression.

```
t1.(5);;
```

```
- : int = 6
```

```
t1.(2) <- 42;;
```

```
- : unit = ()
```

```
t1;;
```

```
- : int array = [|1; 2; 42; 4; 5; 6|]
```

```
t1.(6);;
```

Exception: Invalid_argument "index out of bounds".

```
t1.(-1);;
```

Exception: Invalid_argument "index out of bounds".

Pour créer un tableau, on peut utiliser la fonction `Array.make` : `Array.make n x` renvoie un tableau de taille n dans lequel tous les éléments valent x .

```
Array.make 3 'a';;
```

```
- : char array = [|'a'; 'a'; 'a'|]
```

⚠ Les éléments du tableaux sont alors physiquement tous égaux.

```
let tabtab = Array.make 3 [|1;1|];;
```

```
val tabtab : int array array = [| [|1; 1|]; [|1; 1|]; [|1; 1|] |]
```

```
tabtab.(0).(0) <- 1515;;
```

```
- : unit = ()
```

```
tabtab;;
```

```
- : int array array = [| [|1515; 1|]; [|1515; 1|]; [|1515; 1|]|]
```

Pour initialiser une matrice, on utilisera donc la fonction `Array.make_matrix : int -> int -> 'a -> 'a array array`

```
let tabtab = Array.make_matrix 2 3 0;;
```

```
val tabtab : int array array = [| [|0; 0; 0|]; [|0; 0; 0|]|]
```

```
tabtab.(0).(0) <- 1515;;
tabtab;;
```

```
- : unit = ()
```

```
- : int array array = [| [|1515; 0; 0|]; [|0; 0; 0|]|]
```

Enfin, il est possible de créer un tableau avec la fonction `Array.init : int -> (int -> 'a) -> 'a array` en précisant la longueur du tableau une fonction calculant l'élément d'indice i en fonction de i .

```
let tab = Array.init 5 (fun i -> 2*i);;
```

```
val tab : int array = [|0; 2; 4; 6; 8|]
```

Exercices divers

Exercice 1

Prévoir la réponse de l'interpréteur de commandes :

```
let x = ref 0;;
let z = x;;
x := 4;;
```

```
!z;;
```

Exercice 2

Prévoir la réponse de l'interpréteur de commandes :

```
let a = ref 1;;
let f x = x + !a;;
f 3;;
```

```
a := 4;;
f 3;;
```

Exercice 3

- Écrire une fonction `copy` : `'a array -> 'a array` qui prend en argument un tableau et en renvoie une copie. Quelle est sa complexité ?
- Écrire une fonction `mem` : `'a -> 'a array -> bool` testant l'appartenance d'un élément à un tableau. Quelle est sa complexité ?
- Écrire une fonction `map` : `('a -> 'b) -> 'a array -> 'b array` qui prend en argument une fonction `f` de type `'a -> 'b` et un tableau `[|a1; ...; an|]` d'éléments de type `'a` et qui renvoie le tableau `[|f a1; ...; f an|]`.
- Écrire une fonction `exists` : `('a -> bool) -> 'a array -> bool` qui prend en argument une fonction `f` de type `'a -> bool` et un tableau `[|a1; ...; an|]` d'éléments de type `'a` et qui renvoie `true` si au moins un élément du tableau vérifie `f`, `false` sinon.
- Écrire une fonction `for_all` : `('a -> bool) -> 'a array -> bool` qui prend en argument une fonction `f` de type `'a -> bool` et un tableau `[|a1; ...; an|]` d'éléments de type `'a` et qui renvoie `true` si tous les éléments du tableau vérifie `f`, `false` sinon.

Ces fonctions sont déjà implémentées dans le module `Array`.